This software package is no bomber.

B17

Stewart E. Fason
Via La Selva
Palm Beach, FL 33480

B17
ABS Suppliers
Ann Arbor, MI
$22

Recently, I heard of an extraordinary program. I called the bright young man who wrote it and persuaded him to send me a copy to review. Within three days I received it with some comprehensive instructions. Then the fun began: The old computer mastermind closely examined the program, and, unlike many utilities I've examined, found that it is everything it's cracked up to be.

ABS claims that the B17 will allow you to save, verify and load BASIC or machine language programs almost four times faster than an ordinary cassette recorder. Data arrays (string or numeric) can be saved and loaded up to 165 times faster! All of this magic with no aggravation—nearly for free. It works with Level II, 16, 32 or 48K.

I have just spent the weekend using a B17, and it works. In fact, I am so impressed that I paid for my complimentary copy.

Load and Verify

The B17 is both fast and easy to use. Power-up and set memory size to 31600. Have the B17 in cassette ready to load. Type: SYSTEM ENTER B17 ENTER. The asterisk will flash four times, then a series of little arrows will flow showing that something is being input. In about 10 seconds, READY will appear. CLOAD your BASIC program as usual.

Assume you just loaded your Startrek 16K. It takes about four minutes. Now prepare a new tape to record. Type SAVE "STAR" and enter. Note: You must name each program saved. The name can be any six letters or numbers. While you are saving a program, the little arrows flow to your right showing that data is being output. In just over a minute, Star has been saved.

Verify the save by rewinding the tape, typing LOAD? enter. The arrows go the other way, and, if the save was good, the READY appears. I have yet to discover a bad one.

"The old computer mastermind closely examined the program, and, unlike many utilities I've examined, found that it is everything it's cracked up to be."

When you want to load this new, fast Startrek, answer MEMORY SIZE with enter. Type SYSTEM ENTER STAR ENTER. The B17 has appended itself to the BASIC program. You get four asterisks, then the arrows. When the loading is complete and you run the program, the B17 machine language program disappears leaving full memory available. If you want B17 to remain, reserve space at power up (31600). You reserve memory only if you wish to modify and/or save another copy of the BASIC program.

Data Handling

So much for the fun and games. Let's get down to serious business and see what B17 does with data handling. Your Level II manual shows you how to save and input data using the INPUT#-1 and PRINT#-1 statements. The system is easy, but slow.

Here is a sample program which creates 2000 numbers of six and seven digits and saves the whole lot on tape.

```
10 DIM A (2000)
20 FOR I = 1 TO 2000:B = RND (0): A(I) = B:
   NEXT
30 FOR I = 1 TO 2000:PRINT#-1,A(I): NEXT
```

The above does the job very well in about two hours and 20 minutes.

Now, change line 30 to read 30 PUT. Yes, that's all. . . just PUT. Run the new program.

```
1 REM  YOU MAY USE THIS PROGRAM WITH THE FOLLOWING:
2 REM  1. YOUR TRS-8Ø (NOTHING ELSE REQUIRED)
3 REM  2. STRINGY FLOPPY (HARDWARE/SOFTWARE REQUIRED)
4 REM  3. TC-8  (HARDWARE/SOFTWARE REQUIRED)
5 REM  4. B-17  (SOFTWARE REQUIRED)
1Ø CLS
2Ø CLEAR7ØØØ: REM SAVED SPACE FOR STRING VARIABLES
3Ø DEFSTR A,T:REM DEFINES A & T AS STRING VARIABLES
4Ø INPUT"STRING=1  NUMERICAL=2";QD:IFQD=1THEN 6Ø
5Ø CLEAR: REM NEGATES LINE 3Ø IF DATA INPUT TO BE SOLEL
    Y NUMERIC
6Ø DIM A(5ØØ), B(5ØØ): REM LIMITS NUMBER OF DATA ITEMS
7Ø INPUT"HOW MANY ITEMS TO BE SORTED?";N
8Ø E=Ø
9Ø INPUT"CREATE LIST=1 DATA FROM B17=2 TC8=3 S/F=4 TRS=
    5";S
1ØØ IFS=2THEN11Ø ELSE IFS=3THEN12Ø ELSE IFS=4THEN13Ø EL
    SE IFS=5THEN18ØELSEFORI=1TON:GOTO17Ø
11Ø CLS:INPUT"DATA TAPE IN CASSETTE?  PRESS 'ENTER'";EE
    :GET:FORI=1TON:E=E+1:NEXT:GOTO 53Ø : REM B17 INPUT
     ROUTINE
12Ø CLS:INPUT"DATA TAPE READY? PRESS 'ENTER'";EE:OPEN:F
    ORI=1TON:E=E+1:INPUT#1,A(I):PRINTA(I):NEXTI:CLOSE:
    GOTO 53Ø : REM TC-8 INPUT ROUTINE
13Ø INPUT"OPEN FILE # Ø-9";B: REM STRINGY FLOPPY INPUT
    ROUTINE
14Ø @OPENB
15Ø FORI=1TON:@INPUTA(I):E=E+1:PRINTA(I):NEXTI
16Ø @CLOSE:GOTO 53Ø
17Ø  E=E+1:PRINT"ENTER ITEM #  ";I;:INPUTA(I):NEXTI:GOT
    O19Ø: REM NEW LIST FROM KEYBOARD ROUTINE
18Ø CLS:INPUT"DATA TAPE READY?  PRESS 'ENTER'";EE:FORI=
    1TON:E=E+1:INPUT#-1,A(I):PRINTA(I):NEXT: REM TRS-8
    Ø (SLOW) INPUT ROUTINE
19Ø CLS:OUT254,1: REM DELETE IF NO SPEEDUP KIT
2ØØ L=1: REM SORTING ROUTINE BEGINS
21Ø B(L)=N+1
22Ø M=1
23Ø J=B(L):PRINT@415,"SORTING":CLS
24Ø I=M-1
25Ø IF J-M < 3 THEN 45Ø
26Ø M1=INT(RND(Ø)*(J-M))+M
27Ø I=I+1
28Ø IFI=JTHEN 37Ø
29Ø IFA(I)<=A(M1) THEN 27Ø
3ØØ J=J-1
31Ø IFI=J THEN 37Ø
32Ø IFA(J)>=A(M1) THEN 3ØØ
33Ø T=A(I)
34Ø A(I)=A(J)
35Ø A(J)=T
36Ø GOTO 27Ø
37Ø IF I>=M1 THENI=I-1
38Ø IFJ=M1THEN 43Ø
39Ø T=A(I)
4ØØ A(I)=A(M1)
41Ø A(M1)=T
42Ø L=L+1
43Ø B(L)=I
44Ø GOTO 23Ø
45Ø IFJ-M<2THEN 5ØØ
46Ø IFA(M)<A(M+1) THEN 5ØØ
47Ø T=A(M)
48Ø A(M)=A(M+1)
49Ø A(M+1)=T
5ØØ M=B(L)+1
51Ø L=L-1
515 REM END OF SORTING ROUTINE
52Ø IFL>ØTHEN 23Ø
53Ø OUT254,1:CLS
54Ø FORI=1TO(N-G)
55Ø IFI<1ØTHENFG=2
56Ø IFI>9THENFG=1
57Ø PRINTTAB(FG);I;"";A(I)
58Ø Y=Y+1:IFY=15THENPRINT"NEXT PAGE PRESS 'ENTER'";:INP
    UT:Y=Ø
59Ø NEXTI
6ØØ M=N-G
61Ø G=Ø
615 REM SHOWS AMOUNT OF STRING DATA SPACE YOU HAVE AVAI
    LABLE
62Ø IFQD=1THENPRINTFRE(A)"BITS LEFT";
63Ø OUT254,Ø:Y=Ø:PRINT"LIST AGAIN=Ø SAVE TO TC8=77 TO B
    17=88 TO S/F=99 ":INPUT"TRS8Ø=66 DELETE=-1  TO PRI
    NTER=.9 ADD ITEMS ENTER HOW MANY";UU:IFUU=66THEN79
    ØELSEIFUU=ØTHEN 53Ø ELSEIFUU=99THEN 66Ø ELSEIFUU=-
    1THEN 72Ø ELSEIFUU=88THEN 78Ø
64Ø IFUU=.9THEN 71Ø ELSE IF UU=77 THEN 74Ø
65Ø CLS:N=E+UU:FORI=E+1TON:GOTO 17Ø
655 REM STRINGY FLOPPY DATA SAVE ROUTINE
66Ø INPUT"WHICH FILE #(Ø-9) DO YOU WISH TO WRITE ?";C
67Ø @OPENC
68Ø FORI=1TON:@PRINTA(I):NEXTI
69Ø @CLOSE
7ØØ PRINT:PRINTN;" ITEMS HAVE BEEN SAVED ON S/F":INPUT"
    PRESS ENTER TO RETURN TO MENU";VV:GOTO 63Ø
71Ø FORI=1TON:LPRINTA(I):NEXTI:PRINTN;" ITEMS HAVE BEEN
     PRINTED.":CLS:GOTO 63Ø
72Ø CLS:PRINT:PRINT:INPUT"HOW MANY ITEMS DO YOU WISH TO
     DELETE?";X
73Ø FORI=1TOX:INPUT"ENTER ITEM NO.";Q:A(Q)="[":NEXTI:G=
    G+X:E=E-X:GOTO 19Ø
74Ø CLS:INPUT"READY TO RECORD?  PRESS 'ENTER'";EE:OPEN:
    FORI=1TON:PRINT#1,A(I):NEXTI:PRINTN;" ITEMS HAVE B
    EEN SAVED ON CASSETTE."
75Ø CLOSE
76Ø INPUT"RETURN TO MENU PRESS ENTER";SQ:GOTO 63Ø
77Ø REM B-17 DATA SAVE ROUTINE
78Ø CLS:INPUT"READY TO RECORD?  PRESS 'ENTER'";EE:PUT:I
    NPUT"TO RE-INITIALIZE PROGRAM=1  RETURN TO MENU=2"
    ;SF:IFSF=1THEN1ØELSE 63Ø
785 REM ARCHAIC, SLOW, FRUSTRATING, TRS-8Ø DATA SAVE RO
    UTINE!
79Ø CLS:INPUT"READY TO RECORD?  PRESS 'ENTER'";EE:FORI=
    1TON:PRINT#-1,A(I):NEXT:PRINTN;" ITEMS HAVE BEEN S
    AVED TO TAPE":GOTO63Ø
```

Program Listing 1

(Have your cassette ready to record.) I assume you have B17 in memory. The little arrows dance across the top right corner of your screen for about 20 seconds, and the cassette stops.

There is no way to verify a data save but, using good quality tape, I have had only one bad save in 500. Now, let's try to get the saved data back into memory with this program:

```
10 DIM A (2000)
20 GET
30 FOR I = 1 TO 2000: PRINT A (I): NEXT
```

Your tape has been rewound and set for play. Run the above and watch the arrows flow for about 20 seconds. Then that data will be printed. The B17 GETs and PUTs data faster than the TRS-80 can print it. String data is handled as easily as the above numeric data, however, you must clear the space you need.

The B17 will never replace disks as it will not search for a particular piece of data, however, it has many useful applications.

Program Listing 1 is a natural for the B17. You can use it with a Stringy Floppy or TC-8 as well. This program lets you create a new file, input a previously saved file, add or delete, save or output to your printer. It also sorts—darn fast for BASIC—50 items in about 15 seconds.

In my business I must keep track of over 500 clients. This program does so alphabetically and numerically (by account number). In a matter of minutes my secretary can add, delete, sort and present me with an updated list.

Many of my friends have used this program sucessfully. A group of attorneys use it to keep track of court appearance, filing dates, etc. A series of entries might look like Table 1.

Saving 60-75 percent of the time you spend loading programs makes the B17 well worth the money. Its data handling speed makes it a steal. The introductory price is $22, but by the time you read this, it may be $50 and well worth it.

One note: If you have the cassette mod from Radio Shack, you must add one switch in order to bypass the mod. The B17 won't work with any added audio processing devices.■

00	Judge	Client	Remarks	Attorney
80-09-12	Jones	Smith	File Answer	Hookum
80-09-12	Gray	Ajax Const.	Pre Trial Hearing	Skinnem

Note: The above will printout with the heading on top no matter where it is placed in the program. When the computer sees that first zero and finds no other items beginning with zero, it places that item first. The dates will arrange numerically.

Table 1. Sample Entries for Sorting Program